
17
 TR ADR ; truncate bank address18

19
 XC  ; enable 65C02 code20

21
22

* Internal AppleWorks Equates23
ErrFlag EQU $03 ; ProDOS error occur?24
FoundEsc EQU $88 ; non-zero if <ESC> pressed25
DTCurOpen EQU $C54 ; current file # (if any)26
DTCtOnDesk EQU $C55 ; number of files on current Desktop27
NewStr EQU $D85 ; 128 bytes (used by many routines)28
OrgInMem EQU $E86 ; $00 = in file / $01 = in menu / 29
   ;   $40 = in file list30
StrWork5 EQU $FD7 ; 5-byte pString used by newWord2Str31

32
33

* AppleWorks Host Equates34
AWVersion EQU $1003 ; $33/51 = 5.1 / $28/40 = 4.0  / 35
   ;   $1E/30 = 3.036
PrevSeg EQU $1021 ; SEG in use (last loaded by CallSeg)37
CallSeg EQU $10FA ; AW segment loading routine38
   ;  - LDA with segment # before calling39
   ;     ($01/01 - $2F/47)40
   ;  - see segment chart41
   ;  - add $80/128 to load only (not JMP)42
ClearDA EQU $1100 ; clears display area on screen43
DoGoToXY EQU $1118 ; gotoXY (cursor on screen)44
MvRightRtn EQU $114B ; move in memory routine (positive/right)45
   ; - followed by destination/source/length46
PressAny EQU $1151 ; 'Press space to continue'47
PushStack EQU $1157 ; add a level to Escape road map48
   ; - followed by address of text pString49
StrMvRtn EQU $116C ; copy a pString50
   ; - followed by destination/source51
WritePRtn EQU $1187 ; propagates a character (repeats)52
   ;  (X) times of (Y) character53
DrawBox EQU $11BA ; draws rectangle mousetext box on screen54
   ; - followed by column/row/width/height55

56



57
* AppleWorks SEG $27 (Organizer Main) Equates58
DrawFileCard EQU $211A ; draw AppleWorks FileCard59
   ; - precede with level # in (A)60
SlotClrCurr EQU $2132 ; routine to clear (zero out) contents of61
   ;  "current" file slot from $C56-$C7562
   ; (DTFName/Path/Type/Status/Picked/Size)63
SlotFrCurr EQU $2135 ; routine to save the data about the64
   ;  "current" file back into its65
   ;  proper slot (1-12)66
Slot2Curr EQU $2138 ; routine to load & open one of the67
   ;  desktop files (1-12) and make it the68
   ;  "current" file69
ParmTblPtr EQU $3B28 ; Address of ParmTable Pointer70
RealTypeNew EQU $431E ; 'Type a new name' routine71
PromptNew EQU $432F ; Address of TypeNew Prompt String72
EnterFile EQU $4C35 ; Enter file just loaded73
DTCountRtn EQU $4D1A ; Ensures no more than 12/36 Files74
GoMainEntry EQU $4D3B ; Goto Main Menu and Wait for Input75
NewFrOther EQU $4D99 ; Make WP/Text from 'Other'76
TypeNew EQU $4DBE ; call 'Type a new name' routine77
TryAgain EQU $4DF3 ; Back to Add Files Menu after <ESC> Load78

79
* AppleWorks SEG $28 (File Loader) Equates80
RemFilesRtn EQU $5D25 ; Adds file to Remembered Files81
TextLoader EQU $6366 ; Text Loader in SEG $2882
SetFileStatus EQU $65A1 ; set status of loaded 'Text' file83

84
* AppleWorks SEG $15 (Word Processor Main) Equates85
WPPutLine EQU $6E1A ; stores WpWa/$7B00 work register line 86
   ;  in desktop memory87

88
89

* AppleWorks Subhost Equates90
WriteText EQU $AB37 ; routine to write multiple strings91
   ;  - followed by address of strings 'block'92
   ;  - block contains:   (i) pString for 193
   ;                     (ii) Column for 194
   ;                    (iii) Row for 195
   ;                     (iv) repeat i/ii/iii96
   ;                      (v) $00 = endbyte97
newWord2Str EQU $AB61 ; convert HEX word to decimal98
   ;  (result in StrWork5) [5 character max]99

100
101

* AppleWorks TimeOut Engine Equates102
AccAdr EQU $2100 ; TimeOut Application Loads Here to run103
TOOrg EQU $B80A ; TimeOut Organizer Loaded?104
   ;  $00 = No / $01 = Yes105

106
* TimeOut Edit BASIC Equates107
MoveAdrs EQU $7E00 ; Relocate application here108
   ;  (above SEG $27, SEG $28 and SEG $15)109

110
111
112



************************************************************113
114

 ORG AccAdr ; $2100115
116

Start117
 JMP MainSeg ; skip over TimeOut header118

119
120

*===========================================================121
* Timeout Initialization Header122
*===========================================================123

124
IdBytes HEX 0B02E12644 ; TimeOut unique signature bytes125
MVersion HEX 10 ; Minimum TimeOut version required (1.0)126
MemRes DB 0 ; Memory resident (T/F)127
LockMR DB 0 ; Lock MemRes flag (T/F)128

129
* If your accessory uses any hires graphics, set the following bits in the130
*  'Graphics' byte for correct memory allocation131
Graphics DB 0 ; Bit 0:Hi-Res page 1132
   ;     1:Hi-Res page 2133
   ;     2:Dbl hi-res page 1134
   ;     3:Dbl hi-res page 2135
   ;     4:Super (GS) hi-res136

137
* Maximum 16 character application name138
Name STR 'Edit BASIC 5.4'139
 DS Name-*+17 ; Pad so exactly 17 bytes for name140

141
* The Timeout Engine puts the pathname of the accessory here142
* so that the accessory can reload itself or its segments.143
* NOTE: Move if area will be overwritten144
PathName DS 65145

146
Preserve DS 2 ; These two bytes contain the same value147
   ;  that was left here the previous time this148
   ;  accessory was called.149
ConfigSeg DB 2 ; Configuration segment number if applicable150
ID DB $FF ; ID's were assigned by Beagle Bros151
   ;  (if not assigned = $FF)152
Version DB 54 ;  Version 5.4153
 DS 2154
Reserved DS 9 ; Must be 0's155
NumSegs DB 2 ; Number of segments >= 1156
Offset1 ADR $000000 ; Offset of Segment 1 from Start of File157
Ptr1 DA $0000 ; Filled in by Timeout if memory resident158
Offset2 ADR Config-Start ; Offset of Segment 2 from Start of File159
Ptr2 DA $0000 ; Filled in by TimeOut if memory resident160
EOF DA CodeEnd-Start ; Length of complete TimeOut161
 DA $0000 ; Header END bytes162

163
*---------------------------------------------------------164
* Main Segment165
*---------------------------------------------------------166

167
MainSeg168



 JSR ClearDA ; clears display area on screen169
170

* Version Check for AppleWorks 5.1171
 LDA AWVersion ; version # of this AppleWorks172
 CMP #$33 ; 5.1173
 BCC NoBox ; not version 5.1 - skip box174

175
* Passed Version Check - Can Draw Box176
 JSR DrawBox ; AppleWorks 5.1 routine177
 DB #$0A ; column (10)178
 DB #$03 ; row (3)179
 DB #$3A ; width (58)180
 DB #$0B ; height (11)181

182
* Write Text for Splash Screen183
NoBox LDX #$14 ; column (20)184
 LDY #$06 ; row (6)185
 JSR DoGoToXY ; gotoXY (cursor on screen)186
 LDX #$28 ; width (40)187
 LDY #$CC ; MouseText Top Bar character188
 JSR WritePRtn ; draw MT Top Bar on screen at cursor189

190
 JSR WriteText ; routine to write multiple strings191
 DA SplashScreenA ; address of strings block192

193
* Version Check for AppleWorks 5.1194
 LDA AWVersion ; version # of this AppleWorks195
 CMP #$33 ; 5.1196
 BCC NoGo ; not version 5.1 - message and Quit197

198
* Passed Version Check - Finish Splash Screen & Continue with Application199
 JSR WriteText ; routine to write multiple strings200
 DA SplashScreenB ; address of strings block201

202
 JSR PressAny ; 'Press space to continue'203
 LDA FoundEsc ; <ESC> pressed?204
 BNE WeDone ; escaped out - Quit205
 JMP Begin ; Let's start Application206

207
* Failed Version Check - Finish Alternate Splash Screen & Quit208
NoGo JSR WriteText ; routine to write multiple strings209
 DA SplashScreenC ; address of strings block210

211
 JSR PressAny ; 'Press space to continue'212
 BRA WeDone2 ; Quit (can't reset TOOrg for < 5.1)213

214
* Reset TimeOut Status215
WeDone LDA #$00 ; for noting proper exit of216
   ;  TimeOut App - is set to #$01 when217
   ;   TimeOut organizer is loaded218
 STA TOOrg ; TimeOut Organizer Loaded?219

220
* Quit and Return to AppleWorks221
WeDone2 LDA #$00 ; force re-load of segments -222
 STA OrgInMem ; to ensure patches aren't re-used223
 STA PrevSeg ; to ensure patches aren't re-used224



225
 RTS  ; leave TimeOut and go back to AplWorks226

227
*---------------------------------------------------------228

229
* Common Text230
SplashScreenA STR '***    TimeOut Edit BASIC    ***'231
 DB #$FF ; centered L-R232
 DB #$04 ; row (4)233
 STR '- An AppleSoft BASIC Loader, Viewer and Editor -'234
 DB #$FF ; centered L-R235
 DB #$07 ; row (7)236
 STR 'for AppleWorks 5.1'237
 DB #$FF ; centered L-R238
 DB #$08 ; row (8)239
 STR 'by Hugh Hood'240
 DB #$FF ; centered L-R241
 DB #$0A ; row (10)242
 STR 'version 5.4 (2019)'243
 DB #$FF ; centered L-R244
 DB #$0B ; row (11)245
 STR 'Application Notes'246
 DB #$FF ; centered L-R247
 DB #$0F ; row (15)248
 STR "SSSSSSSSSSSSSSSSS" ; MouseText Mid Bar249
 DB #$FF ; centered L-R250
 DB #$10 ; row (16)251
 DB #$00 ; end byte252

253
* Primary Text when Version Check Passes254
SplashScreenB STR 'o A BASIC program may be saved as a TEXT file. '255
 DB #$FF ; centered L-R256
 DB #$11 ; row (17)257
 STR 'o To enter the saved TEXT file into memory, Quit'258
 DB #$FF ; centered L-R259
 DB #$12 ; row (18)260
 STR 'AppleWorks, and from the BASIC "]" Prompt,'261
 DB #$14 ; column (20)262
 DB #$13 ; row (19)263
 STR '"EXEC" the saved TEXT file.'264
 DB #$14 ; column (20)265
 DB #$14 ; row (20)266
 DB #$00 ; end byte267
268
* Alternate Text when Version Check Fails269
SplashScreenC STR '**  This TimeOut Application REQUIRES AppleWorks 5.1  **'270
 DB #$FF ; centered L-R271
 DB #$13 ; row (19)272
 DB #$00 ; end byte273

274
*---------------------------------------------------------275
* Continue with Application276
*---------------------------------------------------------277

278
Begin279
 JSR MvRightRtn ; Relocate application above280



   ;  SEG $27 (Organizer Main/$2100+), 281
   ;  SEG $28 (File Loader/$5200+), and282
   ;  SEG $15 (WP Main/$6E00+)283
 DA MoveAdrs ; Destination / $7E00284
 DA CodeStart ; Source285
 DA MainCodeEnd-CodeStart ; Length286

287
 JMP MoveAdrs ; Start Application at 288
   ;  Working Address289

290
************************************************************291
CodeStart292
 ORG MoveAdrs ; $7E00293

294
 LDA #$00 ; force load of Organizer Main segment295
 STA OrgInMem ; $00 = in file / $01 = in menu / 296
   ;   $40 = in file list297

298
 LDA #$27+$80 ; Load, don't call Organizer Main299
 JSR CallSeg ; AW segment loading routine300
 LDA #$28+$80 ; Load, don't call File Loader301
 JSR CallSeg ; AW segment loading routine302

303
* Clear out any "current" file to make room for new BASIC file304
 LDA DTCurOpen ; "current" file # (if any)305
 JSR SlotFrCurr ; routine to save the data about the306
   ;  "current" file back into its307
   ;  proper slot (1-12)308
 LDA #$00 ; set "current" file # to equal 0309
 STA DTCurOpen ; Signal no "current" file yet310
 JSR SlotClrCurr ; routine to clear (zero out) contents of311
   ;  "current" file slot from $C56-$C75312
   ; (DTFName/Path/Type/Status/Picked/Size)313

314
* Draw First (2) Filecards in file selection dialog315
 JSR ClearDA ; clears display area on screen316
 LDA #$01 ; file card level #1317
 JSR DrawFileCard ; draw AppleWorks FileCard318
   ; 'Edit BASIC 5.4'already pushed on319
   ;  Stack by TimeOut320
 JSR PushStack ; add a level to Escape road map321
 DA  Prompt1 ; 'Select File'322
 LDA #$02 ; file card level #2323
 JSR DrawFileCard ; draw AppleWorks FileCard324
 JSR PushStack ; add a level to Escape road map325
 DA  PromptQ ; 'Quit'326
   ; NOTE: No need to draw file card level 327
   ;        #3 - it will be drawn by FCList328
   ;        routine called from SEG $28329
   ;        by AskPathName routine330

331
*---------------------------------------------------------------332
* Patches to SEG $27 (Organizer Main) and SEG $28 (File Loader)333
*---------------------------------------------------------------334

335
 LDA #Prompt2 ; Patch Card #3 Stack Msg336



 STA TextLoader+3 ; @ $6369  (LB)337
 LDA #>Prompt2 ; 'BASIC files' 338
 STA TextLoader+4 ; @ $636A (HB)339

340
341

 LDA #ParmTable ; Patch AskPathName ParmTable Address342
   ;  to limited selection to BASIC files343
 STA ParmTblPtr ; $3B28  (LB)344
 LDA #>ParmTable345
 STA ParmTblPtr+1 ; $3B29 (HB)346

347
 LDA #Rename ; Patch 'Type a new name' routine to348
   ;  default to 'FILENAME.T'349
 STA TypeNew+1 ; $4DBF (LB)350
 LDA #>Rename351
 STA TypeNew+2 ; $4DC0 (HB)352

353
 LDA #ChangeName ; Patch address to new prompt string354
   ;  for TypeNew routine355
 STA PromptNew ; $432F (LB)356
 LDA #>ChangeName357
 STA PromptNew+1 ; $4330 (HB)358

359
* Replace/Bypass Remembered Files Routine and Show File as 'New'360
 LDA #$A9 ; LDA Instruction361
 STA RemFilesRtn ; ($5D25)362
 LDA #$01 ; 'New'363
 STA RemFilesRtn+1 ; ($5D25+1)364
 LDA #$8D ; STA Instruction365
 STA SetFileStatus ; Change STZ $0C6C to STA #$01 ('New')366
 LDA #$60 ; RTS / Don't remember this file367
 STA RemFilesRtn+2 ; ($5D25+2)368

369
* Set additional RTS points370
 STA $4C3B ; To Return to TimeOut (instead of371
   ;  MainEntry after sucessful file372
   ;     load and CloseAFile)373

374
 STA $4D4C ; To Return to TimeOut (instead of375
   ;  AddMenu after JSR to DTCountRtn376
   ;  ($4D1A) from this TimeOut App and377
   ;  # of files is < 36)378

379
 STA TryAgain+3 ; $4DF6380
   ; To Return to TimeOut (instead of381
   ;  (Rename after <ESC> Abort) after 382
   ;    CloseAFile patched in below)383

384
* Patch in Return to TimeOut if we already have 36 files385
 LDA #PopAndWrap ; JMP (not RTS) to PopAndWrap386
 STA GoMainEntry+1 ; $4D3C387
 LDA #>PopAndWrap ; 388
 STA GoMainEntry+2 ; $4D3D389
   ;  (instead of going to MainEntry after390
   ;    JSR to DTCountRtn ($4D1A) from391
   ;     this TimeOut App and # of files392



   ;     is 36)393
   ;  Pop RTS because there is no need to 394
   ;   return and JSR to NewFrOther ($4D99)395
   ;   File Loader if there is no room 396
   ;   - already 36 files.397

398
* Patch in CloseAFile Routine before RTS to TimeOut after a Rename <ESC>399
 LDA #$20 ; JSR Opcode400
 STA TryAgain ; $4DF3401
 LDA #$11 ; $2111 (LB) - CloseAFile402
 STA TryAgain+1 ; $4DF4403
 LDA #$21 ; $2111 (HB) - CloseAFile404
 STA TryAgain+2 ; $4DF5405

406
* Patch out direct entry into file after loading407
 LDA #$EA ; NOP Opcode408
 STA EnterFile ; $4C35409
 STA EnterFile+1 ; $4C36410
 STA EnterFile+2 ; $4C37411
   ; Defeat entry of file until412
   ;  AFTER TimeOut is back in413
   ;  control and is finished414
   ;  (normally #$88 Resume Application415
   ;    Call Code for SEG $15 - WP)416
   ;  (Routine is JMPed to from $4DDC and417
   ;   includes CloseAFile routine)418
  419

420
* Fix AppleWorks 5.1 bug when zero files are selectable in FCList421
*   if making a new file from a Text (or any type) file422
*   (should NOT show 'errors loading file' msg if no file was selected)423
 LDA #$EA ; NOP Opcode424
 STA $3B32 ; Was BEQ Opcode in425
 STA $3B33 ;  AskPathName routine426

427
* Patch in code to add printer option to indent long lines428
 LDA #$4C ; JMP to our code429
 STA $645B ; 430
 LDA #AddIndent ; 431
 STA $645B+1 ; 432
 LDA #>AddIndent ; 433
 STA $645B+2 ;434

435
436

* Patch in BASIC parser conversion code437
 LDA #$4C ; JMP to our code438
 STA $6473 ; 439
 LDA #Parse ; 440
 STA $6473+1 ; 441
 LDA #>Parse ; 442
 STA $6473+2 ;443

444
* Patch in code to handle conversion of single to multiple characters445
 LDA #$4C ; JMP to our code446
 STA $6469 ; 447
 LDA #NewReadMore ; 448



 STA $6469+1 ; 449
 LDA #>NewReadMore ; 450
 STA $6469+2 ;451

452
*---------------------------------------------------------------------453
* End of Patches to SEG $27 (Organizer Main) and SEG $28 (File Loader)454
*---------------------------------------------------------------------455

456
 JSR DTCountRtn ; ensures no more than 12/36 Files457

458
 JSR NewFrOther ; ($4D99) Make WP/Text from 'Other'459

460
 LDA ErrFlag ; Any ProDOS Loading Errors?461
 BNE WrapUp ;  (don't enter any file)462

463
 LDA $A57 ; No file was loaded - <ESC> Abort464
   ;  (temporary file load status flag)465
 BEQ WrapUp ;  (don't enter any file)466

467
* Make the last file loaded the current file468
 LDA DTCtOnDesk ; # of files on Desktop469
 STA DTCurOpen ; # of current file470

471
 LDA DTCurOpen ; # of current file472
 JSR Slot2Curr ; bring last file into current area473

474
* Reset TimeOut Status475
WrapUp LDA #$00 ; for noting proper exit of476
   ;  TimeOut App - is set to #$01 when477
   ;   TimeOut organizer is loaded478
 STA TOOrg ; TimeOut Organizer Loaded?479

480
* Quit and Return to AppleWorks481
 LDA #$00 ; force fresh re-load of patched482
 STA OrgInMem ;  segments the next time they are483
 STA PrevSeg ;  called484

485
 RTS  ; leave TimeOut and go back to AplWorks486

487
*-----------------------------------------------------------488
* Remove (2) RTS Before Quitting489
PopAndWrap PLA  ; pop RTS490
 PLA  ; pop RTS491
 BRA WrapUp ; back to AppleWorks492

493
*-----------------------------------------------------------494

495
* Various prompt pStrings496
Prompt1 STR 'Select File'497
Prompt2 STR 'BASIC Files'498
PromptQ STR 'Quit'499
FCLPrmpt1 STR 'Use Return to select a BASIC file or TAB to change Drives'500
FCLPrmpt2 STR 'BASIC file'501
ChangeName STR 'Type an editing name for this BASIC file:  '502

503
*-----------------------------------------------------------504



505
* Parameters for AskPathname/FCList Routine506
ParmTable DA   FCLPrmpt1 ; 'Use Return to select BASIC...'507
 DA   FCLPrmpt2 ; BASIC file508
 DB   $C1 ; Pick (1) Only509
 DB   $FC ; BASIC - FileType for custom description510
 DA   FCLPrmpt2 ; 'BASIC file' - custom description511
 DB   $02 ; (2) Types of files to show512
 DB   $FC ; show BASIC files in list513
 DB   $0F ; show Directory files in list514
 DB   $00 ; End Byte515

516
*-----------------------------------------------------------517

518
* Routine to suggest a '.T' be added to end of BASIC file just loaded519
Rename LDX NewStr ; Length of File Name520
 CPX #$0E ; Less than 14 characters?521
 BCC :RN522

523
 LDX #$0D ; 13 characters524
 STX NewStr525
:RN INX  ; add 1 for '.'526
 LDA #'.' ; '.'527
 STA NewStr,X528
 INX  ; add 1 or 'T'529
 LDA #'T' ; 'T'530
 STA NewStr,X531
 STX NewStr532
 JMP RealTypeNew ; ($431E) / Back to regular rename533

534
*-----------------------------------------------------------535
* BASIC PARSING CODE536
*-----------------------------------------------------------537

538
*-----------------------------------------------------------539
* First Convert Line #s / Ignore #s 0-1 - Convert #s 2-3540
*-----------------------------------------------------------541

542
Parse LDY InputCount ; is Zero on initial entry543
 INC InputCount ; for next pass through544
 CPY #$03 ; #s 0-1 are line pointers545
   ;  and # 2 is Line # Low Byte546
 BCC PreNext ; Ignore; get another547

548
 CPY #$04 ; #s 3 is Line # High Byte549
 BCS CheckRest ; if #4-end is NOT a line number550

551
 STA $65F6 ; Setup for newWord2Str552
 JSR newWord2Str ; convert Hex to Decimal Str553
 DA $65F5 ; Line # in Decimal Str554
   ;  ($65F5 contains 'Y' = $02555
   ;    since it is previous character;556
   ;   $65F6 contains 'Y' = $03557
   ;    since it was read this pass)558

559
 JSR StrMvRtn ; Move converted line # string560



 DA StrWork9 ; TO561
 DA StrWork5 ; FROM562
 INC StrWork9 ; Add 1 for trailing 'space'563
 LDX StrWork9 ; Length of Line # Str + 'space'564
 LDA #' ' ; trailing 'space'565
 STA StrWork9,X ; store trailing 'space'566
567
* Pass multiple characters through on single input character568
StartLoop INC MoreFlag ; Set MoreFlag for multiple chars569

570
 STX CountFlag ; Length of String to Priint571

572
 LDA #$00 ; Initialize Character Counter573
 STA StringCount574
LoopPrint LDX StringCount575
 INX  ; Inc 'X' from 1 thru end576
 LDA StrWork9,X ; Load Character577
 STX StringCount ; Save for next pass578
 CPX CountFlag ; Last Character?579
 BCC Proceed ; No - Just Pass Character Thru580

581
 DEC MoreFlag ; Yes - Unset flag to stop loop &582
 BRA Proceed ;  then Pass Character Thru583

584
*-----------------------------------------------------------585
* Then Convert Tokens, Pass Characters, Check EOL586
*-----------------------------------------------------------587

588
589

CheckRest CMP #$00 ; Is char read (in 'A') a null?590
 BEQ CheckEOL ; Is it EOL?591
 BMI Detoken ; Is it a Token? (>= $80)592

593
 LDX ControlFlag ; show as inverse (config option)594
 BEQ Proceed ; don't Inverse Controls, Just Pass Thru595

596
 CMP #$20 ; Is it Control?597
 BCC ProceedCntrl ; Pass it as Inverse598

599
 BRA Proceed ; NOT EOL or Token or Cntrl - Just Pass Thru600

601
* Eliminate Nulls at End of File Read602
CheckEOL LDA $65F5 ; Previous Byte Read603
 BEQ Next ; Ignore; get another604

605
* Reset InputCount and Issue Carriage Return for EOL606
 STZ InputCount ; begin next line parse607
 JMP $6533 ; Carriage Return Handler608

609
* Trap to discard characters after last EOL610
PreNext CPY #$02 ; Is this the third byte on line?611
 BNE Next ; No - go get another612
 LDA $65F5 ; Check Previous Byte Read613
 BNE Next ; Line Ptr/second byte was NOT zero614
 STZ InputCount ; reset to indicate new line and 615
   ;  disregard altogether616



617
Next JMP $6469 ; Ignore; get another char618

619
620

Proceed JMP $647B ; Back to standard AplWorks filter621
622

ProceedCntrl ORA #$80 ; Make Control Inverse623
 JMP $64A5 ; Pass Control as Inverse and JMP624
   ;  over other character filters625

626
627

Detoken CMP #$EB ; Invalid Token?628
 BCS Next ; Ignore; get another629
 SEC  ; Prepare for subtraction630
 SBC #$7F ; Convert Token to Index #631
 ASL  ; Multiply by (2)632
 CLC633
 TAX  ; Index into TokenStrs Table634

635
 LDA TokenStrs-2,X636
 STA TokenAddress637
 LDA TokenStrs-1,X638
 STA TokenAddress+1639
 JSR StrMvRtn640
 DA StrWork9 ; TO641
TokenAddress DA TokenHOME ; FROM (default string)642
 LDX StrWork9643
 BRA StartLoop644
645
*-----------------------------------------------------------646
* Counters, Flags and Work Strings647
*-----------------------------------------------------------648

649
InputCount DB $00 ; Position on input line650

651
MoreFlag DB $00 ; More Characters to Pass652

653
CountFlag DB $00 ; # of this Character to Pass654

655
StrWork9 DS $0A ; pString to Pass (10 characters)656

657
StringCount DB $00 ; must save and restore on loops658

659
660

*-----------------------------------------------------------661
* CONFIGURATION SEG PARAMETERS STORED HERE662
*-----------------------------------------------------------663

664
ControlFlag DB $01 ; show controls (config item - Yes/No)665

666
IndentCount DB $06 ; default to 6 character Indent667
   ;   (config item - choose 0 - 8)668

669
*-----------------------------------------------------------670
* Handle Instances where (1) Input Character translates to Multiple Characters671
*  (patched into AppleWorks at $6469)672



*-----------------------------------------------------------673
674

NewReadMore LDA MoreFlag ; 675
 BEQ GetNext ; no more to print in the string676

677
 JMP LoopPrint ; Finish rest of string678

679
680

GetNext JSR $63C9 ; Process Next Character Read681
 JMP $646C ; Back to Original AplWorks Code682

683
*-----------------------------------------------------------684
* Allow Indentation of lines that span more than one line.685
*  (patched into AppleWorks at $645B)686
*-----------------------------------------------------------687

688
AddIndent JSR $638B ; finish original $645B code689
 LDA #$DE ; Printer token for indent690
 STA $B5 ; DLLength - holds printer token691
 LDA IndentCount ; # of spaces to indent (config item)692
 STA $B4 ; DLStickies - holds printer token value693
 JSR WPPutLine ; $6E1A - Add Line to Document694
 JSR $638B ; increase line pointer695
 JMP $645E ; back to original code696

697
*===========================================================698
* Address table for token pStrings699

700
TokenStrs DA TokenEND701
 DA TokenFOR702
 DA TokenNEXT703
 DA TokenDATA704
 DA TokenINPUT705
 DA TokenDEL706
 DA TokenDIM707
 DA TokenREAD708
 DA TokenGR709
 DA TokenTEXT710
 DA TokenPR711
 DA TokenIN712
 DA TokenCALL713
 DA TokenPLOT714
 DA TokenHLIN715
 DA TokenVLIN716
 DA TokenHGR2717
 DA TokenHGR718
 DA TokenHCOLOR719
 DA TokenHPLOT720
 DA TokenDRAW721
 DA TokenXDRAW722
 DA TokenHTAB723
 DA TokenHOME724
 DA TokenROT725
 DA TokenSCALE726
 DA TokenSHLOAD727
 DA TokenTRACE728



 DA TokenNOTRACE 729
 DA TokenNORMAL730
 DA TokenINVERSE 731
 DA TokenFLASH732
 DA TokenCOLOR733
 DA TokenPOP734
 DA TokenVTAB735
 DA TokenHIMEM:736
 DA TokenLOMEM:737
 DA TokenONERR738
 DA TokenRESUME739
 DA TokenRECALL740
 DA TokenSTORE741
 DA TokenSPEED742
 DA TokenLET743
 DA TokenGOTO744
 DA TokenRUN745
 DA TokenIF746
 DA TokenRESTORE747
 DA TokenAmp748
 DA TokenGOSUB749
 DA TokenRETURN750
 DA TokenREM751
 DA TokenSTOP752
 DA TokenON753
 DA TokenWAIT754
 DA TokenLOAD755
 DA TokenSAVE756
 DA TokenDEF757
 DA TokenPOKE758
 DA TokenPRINT759
 DA TokenCONT760
 DA TokenLIST761
 DA TokenCLEAR762
 DA TokenGET763
 DA TokenNEW764
 DA TokenTAB765
 DA TokenTO766
 DA TokenFN767
 DA TokenSPC768
 DA TokenTHEN769
 DA TokenAT770
 DA TokenNOT771
 DA TokenSTEP772
 DA TokenPlus773
 DA TokenMinus774
 DA TokenStar775
 DA TokenBSlash776
 DA Token^777
 DA TokenAND778
 DA TokenOR779
 DA TokenGT780
 DA TokenEQ781
 DA TokenLT782
 DA TokenSGN783
 DA TokenINT784



 DA TokenABS785
 DA TokenUSR786
 DA TokenFRE787
 DA TokenSCRN788
 DA TokenPDL789
 DA TokenPOS790
 DA TokenSQR791
 DA TokenRND792
 DA TokenLOG793
 DA TokenEXP794
 DA TokenCOS795
 DA TokenSIN796
 DA TokenTAN797
 DA TokenATN798
 DA TokenPEEK799
 DA TokenLEN800
 DA TokenSTR801
 DA TokenVAL802
 DA TokenASC803
 DA TokenCHR804
 DA TokenLEFT805
 DA TokenRIGHT806
 DA TokenMID807

808
*===========================================================809
* Token pStrings810

811
TokenEND STR ' END ' ; $80...128812
TokenFOR STR ' FOR ' ; $81...129813
TokenNEXT STR ' NEXT ' ; $82...130814
TokenDATA STR ' DATA ' ; $83...131815
TokenINPUT STR ' INPUT ' ; $84...132816
TokenDEL STR ' DEL ' ; $85...133817
TokenDIM STR ' DIM ' ; $86...134818
TokenREAD STR ' READ ' ; $87...135819
TokenGR STR ' GR ' ; $88...136820
TokenTEXT STR ' TEXT ' ; $89...137821
TokenPR STR ' PR# ' ; $8A...138822
TokenIN STR ' IN# ' ; $8B...139823
TokenCALL STR ' CALL ' ; $8C...140824
TokenPLOT STR ' PLOT ' ; $8D...141825
TokenHLIN STR ' HLIN ' ; $8E...142826
TokenVLIN STR ' VLIN ' ; $8F...143827
TokenHGR2 STR ' HGR2 ' ; $90...144828
TokenHGR STR ' HGR ' ; $91...145829
TokenHCOLOR STR ' HCOLOR= ' ; $92...146830
TokenHPLOT STR ' HPLOT ' ; $93...147831
TokenDRAW STR ' DRAW ' ; $94...148832
TokenXDRAW STR ' XDRAW ' ; $95...149833
TokenHTAB STR ' HTAB ' ; $96...150834
TokenHOME STR ' HOME ' ; $97...151835
TokenROT STR ' ROT= ' ; $98...152836
TokenSCALE STR ' SCALE= ' ; $99...153837
TokenSHLOAD STR ' SHLOAD ' ; $9A...154838
TokenTRACE STR ' TRACE ' ; $9B...155839
TokenNOTRACE  STR ' NOTRACE ' ; $9C...156840



TokenNORMAL STR ' NORMAL ' ; $9D...157841
TokenINVERSE  STR ' INVERSE ' ; $9E...158842
TokenFLASH STR ' FLASH ' ; $9F...159843
TokenCOLOR STR ' COLOR= ' ; $A0...160844
TokenPOP STR ' POP ' ; $A1...161845
TokenVTAB STR ' VTAB ' ; $A2...162846
TokenHIMEM: STR ' HIMEM: ' ; $A3...163847
TokenLOMEM: STR ' LOMEM: ' ; $A4...164848
TokenONERR STR ' ONERR ' ; $A5...165849
TokenRESUME STR ' RESUME ' ; $A6...166850
TokenRECALL STR ' RECALL ' ; $A7...167851
TokenSTORE STR ' STORE ' ; $A8...168852
TokenSPEED STR ' SPEED= ' ; $A9...169853
TokenLET STR ' LET ' ; $AA...170854
TokenGOTO STR ' GOTO ' ; $AB...171855
TokenRUN STR ' RUN ' ; $AC...172856
TokenIF STR ' IF ' ; $AD...173857
TokenRESTORE  STR ' RESTORE ' ; $AE...174858
TokenAmp STR ' & '  ; $AF...175859
TokenGOSUB STR ' GOSUB ' ; $B0...176860
TokenRETURN STR ' RETURN ' ; $B1...177861
TokenREM STR ' REM ' ; $B2...178862
TokenSTOP STR ' STOP ' ; $B3...179863
TokenON STR ' ON ' ; $B4...180864
TokenWAIT STR ' WAIT ' ; $B5...181865
TokenLOAD STR ' LOAD ' ; $B6...182866
TokenSAVE STR ' SAVE ' ; $B7...183867
TokenDEF STR ' DEF ' ; $B8...184868
TokenPOKE STR ' POKE ' ; $B9...185869
TokenPRINT STR ' PRINT ' ; $BA...186870
TokenCONT STR ' CONT ' ; $BB...187871
TokenLIST STR ' LIST ' ; $BC...188872
TokenCLEAR STR ' CLEAR ' ; $BD...189873
TokenGET STR ' GET ' ; $BE...190874
TokenNEW STR ' NEW ' ; $BF...191875
TokenTAB STR ' TAB( ' ; $C0...192876
TokenTO STR ' TO ' ; $C1...193877
TokenFN STR ' FN ' ; $C2...194878
TokenSPC STR ' SPC( ' ; $C3...195879
TokenTHEN STR ' THEN ' ; $C4...196880
TokenAT STR ' AT ' ; $C5...197881
TokenNOT STR ' NOT ' ; $C6...198882
TokenSTEP STR ' STEP ' ; $C7...199883
TokenPlus STR ' + '  ; $C8...200884
TokenMinus STR ' - '  ; $C9...201885
TokenStar STR ' * '  ; $CA...202886
TokenBSlash STR ' / '  ; $CB...203887
Token^ STR ' ^ '  ; $CC...204888
TokenAND STR ' AND ' ; $CD...205889
TokenOR STR ' OR ' ; $CE...206890
TokenGT STR ' > '  ; $CF...207891
TokenEQ STR ' = '  ; $D0...208892
TokenLT STR ' < '  ; $D1...209893
TokenSGN STR ' SGN ' ; $D2...210894
TokenINT STR ' INT ' ; $D3...211895
TokenABS STR ' ABS ' ; $D4...212896



TokenUSR STR ' USR ' ; $D5...213897
TokenFRE STR ' FRE ' ; $D6...214898
TokenSCRN STR ' SCRN( ' ; $D7...215899
TokenPDL STR ' PDL ' ; $D8...216900
TokenPOS STR ' POS ' ; $D9...217901
TokenSQR STR ' SQR ' ; $DA...218902
TokenRND STR ' RND ' ; $DB...219903
TokenLOG STR ' LOG ' ; $DC...220904
TokenEXP STR ' EXP ' ; $DD...221905
TokenCOS STR ' COS ' ; $DE...222906
TokenSIN STR ' SIN ' ; $DF...223907
TokenTAN STR ' TAN ' ; $E0...224908
TokenATN STR ' ATN ' ; $E1...225909
TokenPEEK STR ' PEEK ' ; $E2...226910
TokenLEN STR ' LEN ' ; $E3...227911
TokenSTR STR ' STR$ ' ; $E4...228912
TokenVAL STR ' VAL ' ; $E5...229913
TokenASC STR ' ASC ' ; $E6...230914
TokenCHR STR ' CHR$ ' ; $E7...231915
TokenLEFT STR ' LEFT$ ' ; $E8...232916
TokenRIGHT STR ' RIGHT$ ' ; $E9...233917
TokenMID STR ' MID$ ' ; $EA...234918

919
 DB $00 ; END OF TOKEN NAME TABLE920

921
*===========================================================922

923
 ORG924

925
MainCodeEnd926

927
928
*---------------------------------------------------------929
* Configuration Segment930
*---------------------------------------------------------931

932
Config933
* Configuration screen strings - Titles934
 STR 'TimeOut Edit BASIC Configuration'935
 DB #$FF ; Column (centered)936
 DB #$06 ; Row #6937
 STR "SSSSSSSSSSSSSSSSSSSSSSSSSSSSSSSS" ; MT Mid Bar938
 DB $FF ; Column (centered)939
 DB $07 ; Row #7940
941
 DB #$00 ; end of screen strings942
943
*-----------------------------------------------------------944
* Configuration Options945

946
:1 STR 'Show Control Characters in INVERSE' ; Menu Item947
 DB $0D ; Column (Question)948
 DB $0B ; Row (Question)949
 DB $40 ; Column (Answer)950
 DB $0B ; Row (Answer)951
 DB $03 ; Length (Answer)952



 DA :2-* ; Bytes to Next Question953
 DB #$05 ; Yes/No Type Question954
 DB #$01 ; SEG #1 to Patch955
 DA ControlFlag-MoveAdrs+CodeStart-Start ; Offset956

957
958

:2 STR '# of Characters to Indent Long Lines' ; Menu Item959
 DB $0D ; Column (Question)960
 DB $0D ; Row (Question)961
 DB $40 ; Column (Answer)962
 DB $0D ; Row (Answer)963
 DB $03 ; Length (Answer)964
 DA :3-*965
 DB #$03 ; Byte/Integer Type Question966
 DB #$01 ; SEG #1 to Patch967
 DA IndentCount-MoveAdrs+CodeStart-Start ; Offset968
 DB $00 ; Minimum accepted number969
 DB $08 ; Maximum accepted number970

971
:3972
 DB #$00 ; End of Config Parameters973
*---------------------------------------------------------974
 ORG975

976
CodeEnd977
 LST off978

979
 TYP $06 ; BIN file (also need $2100 AuxType)980
 SAV TO.EDITBASIC54981

982


